Newsgroup: it.comp.lang.visual-basic
Titolo: errore 75 filecopy

questo è il codice che eseguo
FileCopy App.Path & "\utilita\autonoleggio.lnk", GetDeskTopPath
ma mi restituisce questo errore 75 che dice
errore di accesso al percordo \ file
qualcuno a idea  di cosa si tratta? 

"Nicolas" wrote in message
news:4869423d$0$18155$
4faf...@reader3.news.tin.it
:

Io no, Nicolas, ma un`idea potresti fartela tu eseguendo, per esempio:
Debug.Print App.Path
Debug.Print App.Path & "\utilita\autonoleggio.lnk"
Debug.Print GetDeskTopPath
Stop
FileCopy App.Path & "\utilita\autonoleggio.lnk", GetDeskTopPath
(Facci sapere se e eventualmente come hai risolto. Grazie!)
--
Ciao!
Maurizio Borrelli [Microsoft MVP Office System]
http://www.riolab.org/


il mio codice è questo, ho provato a creare una stringa PERCORSO ed utilizzo
quella per il filecopy ma non cambia nulla sempre errore 75
ho gestito anche il caso in cui copiassero il contenuto del cdrom in una
cartella da qualche parte in quel caso  serve anche lo "\" iniziale
ho provato a mettere il file di origine in c: e anche la destinazione in c: 
non cambia nulla
ho provato i debug da te indicati e i percosi sono corretti.
non capisco!
    Dim percorso
    If Len(App.Path) = 3 Then
        percorso = App.Path & "utilita\autonoleggio.lnk"
    ElseIf Len(App.Path) > 3 Then
        percorso = App.Path & "\utilita\autonoleggio.lnk"
    End If
    FileCopy percorso, GetDeskTopPath

 qualcuno a idea  di cosa si tratta?
Hai provato in questo modo?
FileCopy App.Path & "\utilita\autonoleggio.lnk", GetDeskTopPath  &
"\autonoleggio.lnk"
Se ti dà ancora errore, potrebbe trattarsi di un problema di privilegi.
--
Ciao Fede
By Chicco Production
http://bcp.freetools.it


hai ragione, molto distrattamente non ho specificato il nome del file di 
destinazione.
funziona perfettamente grazie.
il caldo fà brutti scherzi alla lucidità :)

     FileCopy percorso, GetDeskTopPath
Se stai copiando da CD-ROM a HD potrebbe essere che i file copiati 
abbiano mantenuto l`attributo di sola lettura? Cosi` la seconda volta 
che tenti di sovrascrivere il file ottieni appunto l`errore 75.
In tal caso sperimenta:
Option Explicit
Public Sub Main()
On Error GoTo ErrorHandler
Dim strPath         As String
Dim strSrcPathName  As String
Dim strDstPathName  As String
    Stop ' Premi F8 per continuare.
    strPath = App.Path
    strSrcPathName = "G:\LEGGIMI.TXT"
    strDstPathName = GetStdPath(strPath) _
                   & GetFileName(strSrcPathName)
    If FileExists(strDstPathName) Then
      SetAttr strDstPathName, vbNormal
    End If
    FileCopy strSrcPathName, strDstPathName
ExitProcedure:
    Exit Sub
ErrorHandler:
    With Err
      MsgBox "ERR#" & CStr(.Number) _
           & vbNewLine & .Description
    End With
    Resume ExitProcedure
End Sub
Public Function FileExists(ByVal PathName As String) As Boolean
    FileExists = Len(Dir$(PathName _
                        , vbNormal _
                       Or vbHidden Or vbReadOnly Or vbSystem))
End Function
Public Function FolderExists(ByVal Path) As Boolean
    FolderExists = Len(Dir$(Path _
                          , vbDirectory Or vbNormal _
                         Or vbHidden Or vbReadOnly Or vbSystem))
End Function
Public Function GetFileName(ByVal PathName As String) As String
Dim lngPos  As Long
Dim strRet  As String
    strRet = PathName
    lngPos = InStrRev(strRet, "\")
    If lngPos Then
      strRet = Mid$(strRet, lngPos + 1)
    End If
    GetFileName = strRet
End Function
Public Function GetStdPath(ByVal Path _
                             As String) As String
Dim strRet As String
    strRet = Path
    If Right$(Path, 1) <> "\" Then
      strRet = strRet & "\"
    End If
    GetStdPath = strRet
End Function
